



**HAL**  
open science

# Implementing Type Theory in Higher Order Constraint Logic Programming

Ferruccio Guidi, Claudio Sacerdoti Coen, Enrico Tassi

► **To cite this version:**

Ferruccio Guidi, Claudio Sacerdoti Coen, Enrico Tassi. Implementing Type Theory in Higher Order Constraint Logic Programming. 2017. hal-01410567v2

**HAL Id: hal-01410567**

**<https://inria.hal.science/hal-01410567v2>**

Preprint submitted on 17 Nov 2017 (v2), last revised 6 Nov 2018 (v3)

**HAL** is a multi-disciplinary open access archive for the deposit and dissemination of scientific research documents, whether they are published or not. The documents may come from teaching and research institutions in France or abroad, or from public or private research centers.

L'archive ouverte pluridisciplinaire **HAL**, est destinée au dépôt et à la diffusion de documents scientifiques de niveau recherche, publiés ou non, émanant des établissements d'enseignement et de recherche français ou étrangers, des laboratoires publics ou privés.

# Implementing Type Theory in Higher Order Constraint Logic Programming

Ferruccio Guidi<sup>†</sup> and Claudio Sacerdoti Coen<sup>†</sup> and Enrico Tassi<sup>‡</sup>

<sup>†</sup> *Department of Computer Science and Engineering, University of Bologna*

*ferruccio.guidi@unibo.it, claudio.sacerdoticoen@unibo.it*

<sup>‡</sup> *Université Côte d’Azur, Inria*

*Enrico.Tassi@inria.fr*

*Received 1 July 2017*

In this paper we are interested in high-level programming languages to implement the core components of an interactive theorem prover for a dependently typed language: the kernel — responsible for type-checking closed terms — and the elaborator — that manipulates terms with holes or, equivalently, partial proof terms.

In the first part of the paper we confirm that  $\lambda$ Prolog, the language developed by Miller and Nadathur since the 80s, is extremely suitable for implementing the kernel, even when efficient techniques like reduction machines are employed.

In the second part of the paper we turn our attention to the elaborator and we observe that the eager generative semantics inherited by Prolog makes it impossible to reason by induction over terms containing metavariables. We also conclude that the minimal extension to  $\lambda$ Prolog that allows to do so is the possibility to delay inductive predicates over flexible terms, turning them into (set of) constraints to be propagated according to user provided constraint propagation rules.

Therefore we propose extensions to  $\lambda$ Prolog to declare and manipulate higher order constraints, and we implement the proposed extensions in the ELPI system. Our test case is the implementation of an elaborator for a type theory as a constraint based extension to a kernel written in plain  $\lambda$ Prolog.

## 1. Introduction

### 1.1. A pragmatic reconstruction of $\lambda$ Prolog

In (BBR98) Belleanne et. al. propose a pragmatic reconstruction of  $\lambda$ Prolog (MN86; MNPS91; MN12), the Higher Order Logic Programming (HOLP) language introduced by Dale Miller and Gopalan Nadathur in the ’80s. Their conclusion is that  $\lambda$ Prolog can be characterized as the minimal extension of Prolog that allows to program by structural induction on  $\lambda$ -terms. According to their reconstruction, in order to achieve that goal, Prolog needs to be first augmented with  $\lambda$ -abstractions in the term syntax; then types are added to drive full higher-order unification; then universal quantification in goals and  $\eta$ -

equivalence are required to express relations between  $\lambda$ -abstractions and their bodies; and finally implication in goals is needed to allow for definitions of predicates by structural induction.

By means of  $\lambda$ -abstractions in terms,  $\lambda$ Prolog can easily encode all kind of binders without the need to take care of binding representation,  $\alpha$ -conversion, renaming and instantiation. Structural induction over syntax with binders is also made trivial by combining universal quantification and implication following the very same pattern used in Logical Frameworks like LF (also called  $\lambda$ P). Indeed, LF endowed with an HOLP semantics, is just a sub-language of  $\lambda$ Prolog.

The “hello world” example of  $\lambda$ Prolog is therefore the following two lines program to compute the simple type of a closed  $\lambda$ -expression:

```

kind term, typ type.           type arr typ -> typ -> typ.
type app term -> term -> term. type lam typ -> (term -> term) -> term.

type of term -> typ -> o.
of (app M N) B :- of M (arr A B), of N A.
of (lam A F) (arr A B) :- pi x \ of x A => of (F x) B.

```

## 1.2. $\lambda$ Prolog for proof-checking

According to the Curry-Howard isomorphism, the program above can also be interpreted as a proof-checker for minimal propositional logic. By escalating the encoded  $\lambda$ -calculus to more complex terms and types, it is possible to obtain a proof-checker for a richer logic, like the Calculus of Inductive Constructions (CIC) (PM96; Wer94; Bar99) that, up to some variations, is the common type-theory shared by the interactive theorem provers (ITPs) Coq (dt17), Matita (ARSCT11) and Lean (dMKA<sup>+</sup>15).

All the ITPs mentioned above are implemented following basically the same architecture. At the core of the system there is the *kernel*, that is the trusted code base (together with the compiler and run-time of the programming language the system is written on). The kernel just implements the type-checker together with all the judgements required for type-checking, namely: well-formation of contexts and environments, substitution, reduction, convertibility. The last three judgements are necessary because the type system has dependent types and therefore types need to be compared up to computation.

Realistic implementations of kernels employ complex techniques to improve the performance, like reduction machines to speed up reduction and heuristics to completely avoid reduction during conversion when possible. Is  $\lambda$ Prolog suitable to implement such techniques? The question is not totally trivial because, for example, reduction machines typically rely on the representation of terms via De Bruijn indexes.

We already gave a positive answer in (DGSCT15) where we describe ELPI, a fast  $\lambda$ Prolog interpreter, and implement a type-checker for a dependently type lambda calculus. ELPI is able to run such type checker on the Automath’s proof terms of Landau’s Grundlagen reasonably fast (5 times slower than a corresponding OCaml implementation). In Section 2 we confirm the answer again by implementing a kernel for a generic Pure Type System (PTS) that supports cumulativity between sorts. We then instantiate

the PTS to the one of Matita and we modularly add to the kernel support for globally defined constants and primitive inductive types, following the variant of CIC of Matita. Finally, we embed ELPI in Matita and divert all calls to the  $\lambda$ Prolog kernel in order to test it on the arithmetic library of the system.

### 1.3. From proof-checking to interactive proving

The kernel is ultimately responsible for guaranteeing that a proof built using an ITP is fully sound. However, in practice the user never interacts with the kernel and the remaining parts of the system do not depend on the behavior of the kernel. Where the real intelligence of the system lies is instead in the second layer, called *elaborator* or *refiner* (dMAKR15; ARCT12).

In Section 3 we recall what an elaborator does and we explain why the state of the art is not satisfactory, which motivated at the very beginning our interest in using HOLP languages to implement elaborators.

In Section 4 we pose again the same question we posed for the kernel. Is  $\lambda$ Prolog suitable as a very high-level programming language to implement an elaborator? If not, what shall be added? We conclude that  $\lambda$ Prolog is not suitable and, combining and extending existing ideas in the literature, we introduce the first Constraint Programming extension of  $\lambda$ Prolog. We also implement the language extensions in the ELPI system.

In Section 5 we work towards an implementation of an elaborator for CIC written in ELPI, obtained extending modularly a kernel presented in Section 2. The implementation is the first major use case for the language. The final Section 6 contains comparisons with related work and future works.

## 2. A modular kernel for CIC

We now scale the type-checker for simply typed  $\lambda$ -calculus of Section 1 to a type-checker for a generic PTS that also allows cumulativity between universes and dependent products that are covariant in the second argument. Then we instantiate it to obtain the predicative universal fragment of Luo's ECC (Luo89) and the PTS of CIC, and then we modularly extend the type-checker to the whole CIC by adding inductive types as well.

### 2.1. Term representation and type-checking rules

We start identifying syntactically types and terms, adding a new constructor for sorts of the PTS and by refining *arr* to the dependently typed product. `@univ` is a macro (in the syntax of the ELPI interpreter (DGSCT15)) that will be instantiated in the file that implements the PTS, for example with the type of integers.

```
kind term type.
type sort @univ -> term.      type arr term -> (term -> term) -> term.
type app term -> term -> term. type lam term -> (term -> term) -> term.
```

The `of` predicate is made ternary:

```
type of term -> term -> term -> o.
```

It relates a term, its type and a translation (called *elaboration*) of the term that in this section is a recursive copy of it. It will become a real translation in Section 2.2 to add coercive sub typing in the style of (LSX13) to the kernel, and later again in Section 5.5 to allow implicit argument vectors (the dot-dot-dot notation) during elaboration.

Here we focus on the refined rules for application and lambda abstraction.

The types inferred and expected for the argument of an application are meant to be compared up to  $\beta$ -reduction and cumulativity of universes. The `sub` predicate implements this check. The `match_sort` and the `match_arr` predicates are used to check if the weak head normal form of the first argument is respectively a sort or a dependent product. In both cases the sub-terms are returned by the predicate. For example, `(match_arr (arr nat x \ x) A F)` is meant to instantiate `A` with `nat` and `F` with `(x \ x)`.

```

of (sort I) (sort J) (sort I) :- succ I J.

of (app M N) BN (app RM RN) :-
  of M TM RM, match_arr TM A1 Bx, of N A2 RN, sub A2 A1, BN = Bx RN.

of (lam A F) (arr A B) (lam RA RF) :-
  of A SA RA, match_sort SA (sort _),
  (pi x \ of x RA x => of (F x) (B x) (RF x)), of (arr RA B) _ _.

of (arr A Bx) (sort K) (arr RA RB) :-
  of A TA RA, match_sort TA I,
  (pi x \ of x RA x => of (Bx x) (TB x) (RB x), match_sort (TB x) J),
  max I J K.

```

The rules above have the merit of being syntax-directed, never requiring backtracking and always inferring the most general type, in the sense of Luo (Luo89).

We provide an implementation for `sub`, `match_sort` and `match_arr` in Section 2.3 while `succ` and `max` are described in Section 2.4.

## 2.2. Coercive sub typing

The notion of coercive sub typing (LSX13) relates two formal systems  $T$  and  $T[C]$ : System  $T$  does not feature sub typing while  $T[C]$  extends  $T$  with the set of sub typing judgements  $C$ . The authors of (LSX13) show that a term  $t$  well typed in  $T[C]$  can be related to a term  $t'$  well typed in  $T$ , and that  $t'$  can be obtained inserting in  $t$  explicit type casts around the sub terms that, in order to typecheck, required a typing rule in  $C$ . In other words it gives an algorithm to simulate sub typing in  $T$  by elaborating its terms via the insertion of coercions (explicit type casts) during type checking.

```

of (app M N) BN (app RM RKN) :-
  of M TM RM, match_arr TM A1 Bx, of N A2 RN,
  coerce A2 A1 K, of (app K N) A3 RKN, sub A3 A1, BN = Bx RKN.

```

This alternative rule to type-check applications relies on the user-provided, untrusted `coerce` predicate to find a cast `k` from `A2` to `A1`. These are the clauses that the user adds to the library to implement the classic example of coercions relating the type of natural numbers, integers and rational numbers:

```

type coerce term -> term -> term -> o.

```

```

coerce nat int zpos.
coerce int rat (lam int i \ frac i (zpos (successor zero))).
coerce A C F :- coerce A B F1, coerce B C F2, F = (lam A x \ app F2 (app F1 x)).

```

In the literature on coercive sub typing many flavors of coercions have been studied, e.g. between polymorphic types like lists. All kinds of coercions share the structure above, even if the construction of the composed coercion may be a bit more involved.

### 2.3. Reduction and conversion rules: a trivial but inefficient implementation

To implement `sub`, `match_sort` and `match_arr` we first implement weak head reduction in one step, predicate `whd1`, and then its transitive closure `whd*`. The former is trivial because we reuse the  $\beta$ -reduction of  $\lambda$ Prolog for capture avoiding substitution. The predicates `sub` is then defined by levels: to compare two terms, we reduce both to their weak head normal forms via `whd*` and then compare the two heads. If they match, the comparison is called recursively on the sub terms.

```

type whd1, whd* term -> term -> o.

whd1 (app M N) R :- whd* M (lam _ F), R = F N.
whd* A B :- whd1 A A1, !, whd* A1 B.
whd* X X.

type match_sort term -> @univ -> o.
match_sort T I :- whd* T (sort I).

type match_arr term -> term -> (term -> term) -> o.
match_arr T A F :- whd* T (arr A F).

type conv, conv-whd term -> term -> o.
conv A B :- whd* A A1, whd* B B1, conv-whd A1 B1, !.
% fast path + axiom rule for sorts and bound variables
conv-whd X X :- !.
% congruence rules
conv-whd (app M1 N1) (app M2 N2) :- conv M1 M2, conv N1 N2.
conv-whd (lam A1 F1) (lam A2 F2) :- conv A1 A2, pi x \ conv (F1 x) (F2 x).
conv-whd (arr A1 F1) (arr A2 F2) :- conv A1 A2, pi x \ conv (F1 x) (F2 x).

type sub, sub-whd term -> term -> o.
sub A B :- whd* A A1, whd* B B1, sub-whd A1 B1, !.
sub-whd A B :- conv A B, !.
sub-whd (sort I) (sort J) :- lt I J.
sub-whd (arr A1 F1) (arr A2 F2) :- conv A1 A2, pi x \ sub (F1 x) (F2 x).

```

### 2.4. Defining the Pure Type System

To obtain the kernel, the programmer just needs to accumulate together the  $\lambda$ Prolog file that implements the type-checking rules, the one that deals with reduction and conversion, and a final file containing the definition of a particular PTS. The latter file just needs to implement the `succ`, `max` and `lt` predicates over universes.

For example, the following lines define the predicative hierarchy of Luo's ECC (Luo89),

using the integer  $i$  to represent the universe  $\text{Type}_i$ . The `macro` directive is recognized by the ELPI interpreter that transparently replaces all occurrences of `@univ` by `int`.

```
macro @univ :- int.
max N M M :- N =< M.    lt I J :- I < J.
max N M N :- M < N.    succ I J :- J is I + 1.
```

### 2.5. Reduction and conversion rules: an efficient implementation via a reduction machine for call-by-need

The implementation of weak head reduction given before is well-known to be very inefficient, because the body of the abstraction of a  $\beta$ -redex is immediately and completely traversed when the redex is fired. Moreover, the strategy implemented by `whd*` is call-by-name, that is inefficient per se.<sup>†</sup>

To speed up reduction, we now provide a different implementation based on the reduction machine for call-by-need called Wadsworth Abstract Machine (ABM14). The machine is a modified Krivine Abstract Machine (KAM (Kri07)) that records in the machine environment both terms and their normal forms, the latter being computed lazily by a new invocation of the reduction machine that is fired only when the normal form is required for the first time, i.e. when the variable bound to that value occurs in head position during reduction.

In standard implementations, variables are represented by De Bruijn indexes and machine environments are stacks, indexed by the variables. Since we do not use and do not want to use indexes, we take a different approach. The machine environment is represented by a set of `(val N T V NF)` assumptions that we dynamically add to the  $\lambda$ Prolog environment using logical implication. Their meaning is that to the variable `N` we associate a value `V` of type `T` and its normal form `NF`. The latter is initialised with a fresh metavariable when the binding for `N` is destroyed by a  $\beta$ -reduction, and it is filled in when the value is requested for the first time. To compute the term to be aligned to `NF` we first start a new machine on `V` and then we decode the final machine state, an operation that we call *unwind*. We call a variable `N` for which a `(val N _ _)` assumption is present *val-bound*.

Fetching the normal form of `x` from the environment via a call to `(val x _ _ NF)` amounts to fetching a clause from the current program, that is performed efficiently (e.g. via hash-tables) in any reasonable implementation of  $\lambda$ Prolog thanks to clause indexing. ELPI, as most Prolog engines, indexes clauses on the head predicate *and its first argument*: since `x` is a fresh constant there is little risk of having collisions. Even if the indexing ignores the predicate argument or if all keys generate collisions, the cost of the lookup is at worst  $O(n)$  where  $n$  is the size of the reduction machine environment. In line with what one would obtain by representing the environment as a linked list and

<sup>†</sup> Note that  $\beta$ -reduction is not the only rewriting rule of type theories like CIC, that also features abbreviation unfolding and pattern matching. Hence, even if the  $\lambda$ Prolog runtime implements an explicit substitution calculus and performs substitution lazily, it is not sufficient to piggy-back on the programming's language runtime to fully solve the inefficiency.

variables with De Bruijn indexes. In practice, we expect lookup to be constant time in most cases.

An assumption  $A$  introduced in  $\lambda$ Prolog via  $A \Rightarrow B$  is only visible in  $B$ . Therefore, once a variable is bound via an assumption  $A$  of the form  $(\text{val } x \text{ t } n \text{ nf})$ , we need either 1) to be sure that the rest of the computations that requires  $x$  is performed in  $B$ ; or 2) to reintroduce syntactically the binding again around every term that escapes the scope of the assumption.

To achieve 1) we code the reduction machine in Continuation Passing Style (CPS). The `whd1` predicate relates the head and stack of the machine together with two continuations  $\kappa_s$  (for success) and  $\kappa_f$  (for failure).  $\kappa_s$  is eventually called with the new machine head and stack (as well as the list of val-bound variables). In case no reduction step can be performed  $\kappa_f$  is called. The `whd*` predicate is also implemented in CPS style, and it is responsible for iterating `whd1` and composing together the lists of val-bound variables.

The list of val-bound variables is used when reduction is over and we want the final head and stack to escape the CPS style computation. To explicitly bind the context of `val` assumptions we introduce the term constructor for local abbreviations `abbr` (also called `let .. in` in several functional programming languages), and the `val-to-abbr` predicate to perform the conversion. The `whd_unwind t NF` predicate computes the normal form  $NF$  of  $t$  by calling `whd*` on  $t$  and passing a continuation that unwinds the final machine state by introducing the explicit binder  $(\text{abbr } ty \text{ val } n \setminus ..)$  for each val-bound variable  $n$ . The signature and type-checking rule for `abbr` follow:

```
type abbr term -> term -> (term -> term) -> term. % local definition (let-in)

of (abbr TY TE F) (abbr TY TE G) :-
  of TE TY1, sub TY1 TY, pi x \ of x TY => val x TY TE NF => of (F x) (G x).
```

The code of the reduction machine is the following:

```
macro @stack :- list term. macro @val-ctx :- list term.
macro @continuation :- @val-ctx -> term -> @stack -> o.
type whd1 term -> @stack -> @continuation -> o -> o.
type whd* term -> @stack -> @continuation -> o.

% KAM-like rules in CPS style
whd1 (app M N) S Ks Kf :- !, Ks [] M [N|S].
whd1 (lam T F1) [N|NS] Ks Kf :- !, pi x \ val x T N NF => Ks [x] (F1 x) NS.
whd1 X S Ks Kf :- !, val X _ N NF, if (var NF) (whd_unwind N NF), Ks [] NF S.
whd1 X S Ks Kf :- Kf.

% reflexive, transitive closure
whd*-aux T S VL K :-
  whd1 T S (v1\ t1\ s1\ sigma VL2\ append VL v1 VL2, whd*-aux t1 s1 VL2 K)
  (K VL T S).
whd* T S K :- whd*-aux T S [] K.

% bind the val context back
type val-to-abbr list term -> term -> term -> o.
val-to-abbr [] NF NF.
val-to-abbr [X|XS] I (abbr T N K) :- val X T N _, val-to-abbr XS I (K X).

% whd followed by machine unwinding.
```

```

type whd_unwind term -> term -> o.
whd_unwind N NF :-
  whd* N [] (l\ t\ s\ sigma TS\ unwind_stack s t TS, val-to-abbr l TS NF).

% decodes a machine head and stack to a term
type unwind_stack @stack -> term -> term -> o.
unwind_stack [] T T.
unwind_stack [X|XS] T O :- unwind_stack XS (app T X) O.

```

The predicate `match_sort` reduces a term in weak head normal form and asserts it is a `(sort I)` extracting the universe `I`, that is not a term and hence can trivially escape the CPS computation.

The predicate to match a dependent product, `match_arr`, is more delicate since it has to extract two subterms once the input term is put in weak head normal form by `whd*` and the resulting term is checked to be a product. The subterms may contain val-bound variables, hence the continuation of `whd*` has to resort to `val-to-abbr` to close the two terms, similarly to `whd_unwind`.

```

type match_sort term -> @univ -> o.
match_sort T I :- whd* T [] (l\ t\ s\ t = sort I, s = []).

type match_arr term -> term -> (term -> term) -> o.
match_arr T A F :-
  whd* T [] (l\ t\ s\ sigma A1 \ sigma F1\
    s = [], t = arr A1 F1, val-to-abbr l A1 A,
    pi x\ val-to-abbr l (F1 x) (F x)).

```

Efficient reduction is not sufficient to obtain a quick term comparison routine. Indeed both Matita and Coq complement it with a similar heuristic: put the two terms being compared in weak head normal form *step-by-step*, possibly avoiding to reach the normal form. At each step  $\alpha$ -conversion (i.e.  $\lambda$ Prolog equality) is tried along with congruence rules in order to avoid unnecessary reductions.

In order to perform weak head reductions lazily, we obtain `conv` and `sub` as two instances of a `comp` comparison predicate that works on two reduction machine statuses. The third argument is `eq` if the intended comparison is `conv` while `leq` for `sub`.

```

type conv term -> term -> o.
type sub term -> term -> o.
type comp term -> @stack -> eq_or_leq -> term -> @stack -> o.

conv T1 T2 :- comp T1 [] eq T2 [].
sub T1 T2 :- comp T1 [] leq T2 [].

% congruence + heuristic
comp (sort I) [] leq (sort J) [] :- lt I J.
comp T1 S1 _ T1 S1 :- !.
comp X S1 _ X S2 :- forall2 S1 S2 conv, !. % attempt with no head reduction
comp (lam T1 F1) [] eq (lam T2 F2) [] :- conv T1 T2, pi x \ conv (F1 x) (F2 x).
comp (arr T1 F1) [] D (arr T2 F2) [] :- conv T1 T2, pi x \ comp (F1 x) [] D (F2 x) [].
% reduction
comp T1 S1 D T2 S2 :-
  whd1 T1 S1 (_\ t1\ s1\ whd1 T2 S2 (_\ t2\ s2\ comp t1 s1 D t2 s2)
    (comp t1 s1 D T2 S2))
  (whd1 T2 S2 (_\ t2\ s2\ comp T1 S1 D t2 s2) fail).

```

2.5.1. *Coverage of the full type theory of Matita* The type theory of Matita and Coq is an extension of ECC with global definitions, declarations, primitive inductive types, case analysis and structural recursive definitions.

We implemented the type checking rules for all these extensions in a modular way. To activate one extension, it is sufficient to accumulate in the kernel a  $\lambda$ Prolog file that adds new constructors to the `term` and to the reduction machine stack type, and the relative clauses to the `whd1`, `conv` and `of` predicates. We omit the implementation of the extensions in the paper. All together, they provide a kernel that is functionally equivalent to the one of Matita, except for the verification of soundness of inductive type declarations and termination of recursive functions that we have not implemented yet.

To assess the resulting kernel we linked ELPI into Matita and we made the two kernels (the original one of Matita and the one described in this paper) run side by side while type checking the arithmetic library of the system.

According to our measurements the kernel presented in the paper is 42 times slower than the interpreted version of Matita (to be fair versus ELPI that is an interpreter). A previous implementation of a type-checker for Automath was only 5 times slower (DGSC15), suggesting the possibility to optimize the code further. Indeed, we have a second kernel that is entirely written in CPS style (to avoid turning `vals` into `abbrs` back and forth) that is only 9 times slower. We do not present it in the paper because it is harder to turn into an elaborator, which is our goal for the rest of the paper.

The size of the code is remarkably small, begin around 700 lines, including comments and type annotations. The one of Matita, that is one of the smallest implementing such type theory (ARSCT09), is 5 times larger (after excluding the checks not implemented in the other kernel).

### 3. The elaborator component of today's ITPs

An elaborator takes in input a *partial term*, and optionally a type, and returns the closest term similar to the input one such that the term has the expected type. Both the input and output terms are partial in the sense that sub terms can be omitted and replaced with named holes to be later filled in or, in logic programming terminology, with *existentially quantified metavariables*. For example, the partial term  $\lambda x : T. f (X x) Y$  where  $T, X, Y$  are quantified outside the term represents the  $\lambda$ -abstraction over  $x$  of a type yet to be determined of the application of  $f$  to two arguments, both to be yet determined and such that  $x$  can appear free only in the first one. Elaborating the term versus the expected type  $\mathbb{N} \rightarrow \mathbb{B}$  will instantiate  $T$  with  $\mathbb{N}$  and verify that  $f$  is a binary function returning a boolean.

The importance of the elaborator is twofold. On the first hand, it allows to interpret the terms that are input by the user, usually by means of a user-friendly syntax where information can be omitted, mathematical notation is used and abused, sub typing is assumed even if elements of the first type can only be coerced to elements of the second by inserting a function call in the elaborated term. A better elaborator therefore gives to the user the feeling of a more intelligent and user friendly system. On the other hand, via Curry-Howard, a partial term is a partial proof and an ITP is all about instantiating

holes in partial proofs with new partial terms to advance in the proof. The elaborator is thus the mechanism that takes a partial sub-proof and makes it fit in the global one to progress on a particular proof obligation. In other words, all tactics of the ITP ultimately produce partial proof terms that are elaborated. The more advanced is the elaborator, the simpler the code implementing tactics can be.

### 3.1. Implementing an elaborator: state of the art

The elaborators of the majority of interactive provers are implemented according to the same schema: the syntax of terms is augmented with explicitly substituted existential variables and the judgements of the kernel are re-implemented from scratch, generalizing them to take into account metavariables and elaboration.

In particular the elaborator code works on two new data types: one for partial terms and one, called *metasenv*, that assigns to metavariables a typing judgement (a sequent) and, eventually, an assignment. E.g. a *metasenv* containing  $x:\text{nat}, y:\text{bool} \vdash X \times Y : \text{nat}$  declares  $x$  to be an hole to be instantiated only with terms of type `nat` in the context  $x:\text{nat}, y:\text{bool}$ .

All algorithms manipulating terms are extended to partial terms. For example, conversion (the `sub` predicates in the running example) becomes narrowing, i.e. higher order unification under a mixed prefix (Mil92) in presence of rewriting rules (and cumulativity of universes as sub typing). Unification requires metavariable instantiation, that is implemented lazily for efficiency. The type checking predicate `of` is generalized to elaboration, for example by replacing all calls to conversion with calls to narrowing and by threading around the *metasenv*.

The state-of-the-art approach is sub-optimal in many ways:

**programming platform weakness** Much of this new code has very little to do with the prover or the implemented logic: in particular code that deals with binders ( $\alpha$ -conversion, capture avoiding substitution, renaming) and code that deals with existentially quantified metavariables (explicit substitution management, name capturing instantiation).

**intricacy of algorithms** Such code is far from being trivial, since it tackles problems that, like higher order unification, are only semi-decidable. For efficiency reasons a lot of incomplete heuristics are implemented to speed up the system and reduce backtracking. The heuristics are quite ad-hoc and they interact with one another in unpredictable ways. Because they are hidden in the code, the whole system becomes unpredictable to the user.

**code duplication** Given such complexity in the elaborator, and the safety requirements of interactive provers, the kernel of the system is kept simple by making it unaware of partial terms. As a consequence a lot of code is duplicated, and the elaborator ends up being a very complicated *twin brother of the kernel* (Huet's terminology).

**twins' disagreement** Worse than that, such twin components need to agree on ground terms. Typically a proof term is incrementally built by the elaborator: starting from a metavariable that has the type of the conjecture the proof commands make progress

by instantiating it with partial terms. Once there are no unresolved metavariables left, the ground term is checked, again and in its totality, by the kernel.

**extensibility of the elaborator** Finally, the elaborator is the brain of the system, but it is oblivious of the pragmatic ways to use the knowledge in the prover library, e.g. to automatically fill in gaps (GAA<sup>+</sup>13; ARCT09), to coerce data from one type to another (Luo96) or to enrich data to resolve mathematical abuse of notation (CT09). Therefore systems provide ad-hoc extension points to increase the capabilities of the elaborator. The languages to write this code are typically high-level, declarative, and try to hide the intricacies of bound variables, metavariables, etc. to the user. The global algorithm is therefore split in multiple languages, defying the hope for static analysis and documentation of the elaborator.

### 3.2. The proposed approach: the semi-shallow embedding

The motivation of our research is to improve over the state of the art by identifying an high-level, logic, programming language suitable for the implementation of elaborators. In particular:

- 1 The programming language takes care of the representation of *bindings and metavariables* in the spirit of semi-shallow embedding (DCT16), solving the **programming platform weakness** issue. It also improves on **extensibility** by allowing both the core implementors and the users to work on the same, high-level code. Finally the **intricacy of elaboration** is mitigated.
- 2 The programming language features a primitive and powerful notion of extensibility: programs are organised into clauses, and new clauses can be freely added. In this way the rules of the kernel need not to be re-implemented in the elaborator. On the contrary, they are extended to cover partial terms, solving the **code duplication** issue. Also, the **twins' disagreement** problem becomes less severe, since most code is shared, and **extensibility of the elaborator** become less ad-hoc: the user simply declares new clauses.
- 3 Finally the programming language has a clean semantics, making it easy to prove that the extensions to the kernel only accept partial terms that are, once completed, well-typed according to the core set of rules of the kernel. This completely solves the **twins' disagreement** problem, making it possible to merge the kernel and the elaborator.

We envisage such language to be a logic one for two reasons: first we hope to reuse or at least extend  $\lambda$ Prolog; second we observe that another component of each ITP, the one implementing proof commands, can take real advantage from a programming language where backtracking is built-in, in particular to write proof commands performing proof search.

The Higher Order Abstract Syntax approach identifies the object language binders and the meta-language ones, obtaining  $\alpha$ -conversion and  $\beta$ -reduction for free. The semi-shallow embedding (DCT16) we expect to use in our language *identifies the metavariables of the object language with the metavariables of  $\lambda$ Prolog*. Such metavariables already come

in  $\lambda$ Prolog with automatic instantiation, context management and forms of higher order unification.

At a first sight, the runtime of  $\lambda$ Prolog seems to already provide metavariables and all related operations required for the semi-shallow embedding. So does the technique work out of the box? Unfortunately not quite, as we will see in the next section.

#### 4. $\lambda$ Prolog meets partial terms

We already know from (BBR98) that  $\lambda$ Prolog is the minimal extension of Prolog that allows to implement inductive predicates over syntax containing binders. Does it work when applied to data that is meant to contain existentially quantified metavariables too?

##### 4.1. From generative semantics to constraints

Consider the  $\lambda$ -term  $(\mathit{lam} \ a \ \_ \ P \ a)$  that encodes a partial proof of  $\forall A, A \Rightarrow A$ . If we run the following query, the computation diverges:

```
?- of (lam T a \ P a) (arr (sort I) a \ arr a _ \ a) _
```

The rule for  $\lambda$ -abstraction applies fine, but the resulting goal  $(\mathit{of} \ (P \ x) \ (\mathit{arr} \ x \ \_ \ x) \ RP)$ , for some fresh  $x$ , is problematic. Indeed the type checking rule for application, which head is  $\mathit{of} \ (\mathit{app} \ M \ N) \ BN \ (\mathit{app} \ RM \ RN)$ , applies indefinitely to it.

The  $\mathit{of}$  predicate inherits from Prolog a generative semantics: when called recursively on a flexible input, it enumerates all instances trying to find the ones that are well-typed. Even when the computation does not diverge, the behaviour obtained is not the one expected from an elaborator for an *interactive* prover: the elaborator is not meant to fill in the term, unless the choice is obliged. On the contrary, it should leave the metavariable not instantiated and should *remember* (in some kind of metasenv) the need for verifying if the typing judgement holds later on, when the metavariable gets instantiated. In the example above, type-checking  $(\mathit{lam} \ T \ a \ \_ \ P \ a)$  forces the system to remember that a term of type  $(\mathit{arr} \ x \ \_ \ x)$  has to be provided (in a context where  $\mathit{of} \ x \ (\mathit{sort} \ I) \ x$  holds), that in turn corresponds to the proof obligation  $A : \mathit{sort} \ I \vdash A \Rightarrow A$ .

As we mentioned earlier the sometimes undesired generative semantics is inherited from Prolog. Nevertheless, all modern Prolog engines provide a **var**/1 built-in to test/guard predicates against flexible “input”, provide one/many variants of **delay**/2 to suspend a goal until the “input” becomes rigid, and provide modes declarations to both statically/dynamically detect problematic goals and to (semi) automatically suspend them. For example, by using the **delay** pack of SWI-Prolog (WSTL12), the goal `plus(X, 1, Y)` is delayed until either  $x$  or  $y$  are instantiated. Delayed goals can be thought as *constraints* over the metavariables occurring in them.

These mechanisms, however, have never been standardized. Some of them break the clean declarative semantics of Prolog, others respect it.  $\lambda$ Prolog does not provide any of these facilities, or at least, does not expose them to the programmer. For example, the Teyjus system delays unification problems outside  $L_\lambda$  and implements some machinery to wake them up when they re-enter the fragment, but does not expose any primitive to delay other kind of goals.

In the light of all these considerations we extend the ELPI  $\lambda$ Prolog interpreter with a **delay** directive that we can use as follows:

```
delay (of X T Y) on X.
```

The interpreter now delays goals of the form  $(\text{of } X \ T \ Y)$  whenever  $X$  is flexible. Instead of diverging, the running example now terminates leaving the following (suspended) goal unsolved:

```
of x (sort I) x ?- of (P x) (arr x _\ x) (RP x)
```

Such suspended goal is to be seen as a typing *constraint* on assignments to  $P$ : when  $P$  gets instantiated by a term  $t$ , the goal  $(\text{of } (t \ x) \ (\text{arr } x \ _\ x) \ (R \ x))$  is resumed and  $(t \ x)$  is checked to be of type  $(\text{arr } x \ _\ x)$ . In turn such check can either:

**terminate** successfully if  $t$  has the right type and is ground, e.g.  $(t = x \ \text{lam } x \ w \ w)$ . If such assignment for  $P$  comes from a proof command, then it corresponds to a proof step that closes the goal.

**fail** rejecting the proposed assignment for  $P$  when  $t$  is ill-typed or its type is wrong, e.g.

```
 $(t = x \ \text{lam } x \ w \ x)$ . This corresponds to an erroneous proof step.
```

**advance** and generate one or more new constraints if  $t$  is partial e.g.  $(t = x \ \text{lam } x \ w \ Q \ x \ w)$ .

If  $t$  is generated by a proof command, then this corresponds to a proof step that opens one or more new goals.

The three scenarios above perfectly match the desired behavior of an elaborator. In addition to that, the set of delayed goals implicitly kept by the language interpreter represents faithfully the metasenv data structure, relieving the programmer from managing it himself, threading it through all typing rules and restoring an old copy explicitly on backtracking. Moreover, the automatic resumption of typing constraints makes it impossible to obtain an ill-typed term by instantiation: the code is correct with respect to this invariant by construction.

#### 4.2. Constraint propagation as meta-theorems on ground terms

In many situations, the constraints that accumulate over time are not independent, and sometimes sets of constraints can be rewritten in order to simplify them, or detect their unsatisfiability and backtrack. For example, if our metavariable  $P$  does not occur linearly, one can end up with two distinct typing constraints on it:

```
of x (sort I) x ?- of (P x) (arr x _\ x) (RP x)
of z (sort I) z ?- of (P z) (arr (T z) y\ S z y) (RP z)
```

If the object language features uniqueness of typing, as it is the case for CIC (up to universe sub typing), one surely wants to get rid of the second constraint, and force  $(T = a \ \backslash \ a)$  and  $(S = a \ \backslash \ b \ \backslash \ a)$ . This way, when  $P$  gets instantiated *only one goal* is resumed (to type check the assigned term). Alternatively, if the two typing constraints clash, one wants the elaborator to backtrack immediately.

Languages (or libraries for existing languages) that allow constraints declaration usually come with ad-hoc *constraint solvers* that take care of propagating a specific class of

constraints (arithmetic, finite domain, etc...), and are then called *Constraint Programming* (CP) languages. For example, the set  $\{0 \leq N \leq 4, 2 \leq N \leq 5\}$  can be rewritten into  $\{2 \leq N \leq 4\}$  without changing the set of ground solutions.

Most CPs do not allow the user to define new constraints and propagation rules in user space: only constraints belonging to well-known categories can effectively be used. In our case the constraints originate from the meta-theory of the object language we are implementing, hence the programmer must be able to declare new kind of constraints and specify how constraints are to be rewritten according to meta-theorems about the object language (e.g. uniqueness of typing).

Languages to describe sets of rules to manipulate a set of constraints have been studied in the literature, and the most well-known one is certainly CHR (Constraint Handling Rules (Frh09)). CHR is a first-order language in which the user declares predicates and then gives a set of rewriting rules of the form  $S_1 \setminus S_2 \mid G \iff S_3$  where all  $S_i$  are set of constraints and  $G$  is a guard and whose declarative semantics is that  $\bigwedge S_1 \wedge G \Rightarrow (\bigwedge S_2 \iff \bigwedge S_3)$  and whose operational semantics is to match the current constraints against both  $S_1$  and  $S_2$  and, if the clause  $G$  holds, replace the constraints in  $S_2$  with the ones in  $S_3$ . All syntactic components can be omitted:  $G$  defaults to **true** and the sets  $S_1, S_2, S_3$  to the empty set. The  $\iff$  symbol is also omitted when  $S_3$  is, and  $\mid$  is omitted when  $G$  is. The semantics is completed by a strategy that fixes the order in which propagation rules are fired and in what cases propagation rules can be backtracked.

Extending  $\lambda$ Prolog with a **delay** construct (to declare constraints) and a CHR-like language to declare constraint propagation rules enables the following applications:

**forward reasoning during search** When propagation rules are theorems over the ground instances, then propagation respects the declarative, proof theoretic semantics of  $\lambda$ Prolog. In particular, constraint propagation is just a controlled form of forward reasoning, while SLD resolution only does backward reasoning.

**code optimizations given by the meta-theory** of the object language. In the running example, uniqueness of typing is key to keep the implicit metasenv linear in the number of missing sub-proofs. In turn, this greatly reduces the number of checks to be performed when a metavariable is instantiated, and it allows to directly present to the user the set of constraints as the frontier of the proof search.

**turning check into inference** By replacing the predicates **<**, **max**, **succ** on universes with generic order constraints **leq**, **ltn** one gets an elaborator that works with floating sorts (Her05), rather than fixed integers, and maintains an acyclic graph of constraints linking these. More generically, we can turn type checking into type inference and elaboration.

In the light of all these considerations we extend the ELPI  $\lambda$ Prolog interpreter with a higher order constraint propagation language inspired by CHR.

The following propagation rule is valid code, and implements uniqueness of typing:

```
constraint of {
  rule (G1 ?- of X T1 _) \ (G2 ?- of Y T2 _) > X ~ Y
    | (equiv G1 G2) <=> (G1 => conv T1 T2).
}
```

Here  $(G1 \text{ ?- of } X \ T1 \ \_)$  is a constraint required to be present,  $(G2 \text{ ?- of } Y \ T2 \ \_)$  is another one we want to remove. The type of  $G1$  and  $G2$  is  $(list \ o)$ , i.e. it is a first class representation of the  $\lambda$ Prolog context (that augments the original program). The guard  $(equiv \ G1 \ G2)$  is a user defined  $\lambda$ Prolog program checking that the two contexts are equivalent (i.e. contain the same items, disregarding order and multiplicity).  $G1 \Rightarrow conv \ T1 \ T2$  (syntactic sugar for  $H1 \Rightarrow \dots \Rightarrow HN \Rightarrow conv \ T1 \ T2$  where  $G1 = [H1, \dots, HN]$ ) is the new goal generated by the rule. The alignment expressions,  $> X \ \sim \ Y$ , asserts  $X$  and  $Y$  are the same metavariable and finds a bijection between the eigenvariables (introduced by  $\pi$ i rules) in the two goals.

We say that CHR rules run at the *meta-level*, i.e. they have access to the syntax of goals and to their context. In particular, all flexible terms in a constraint are replaced by fresh constants before being inspected by a CHR rule (we call this operation freezing). Therefore, the propagation rules can inspect the goal, compare frozen metavariables, or detect frozen flexible terms etc. without triggering the instantiation of the metavariable. As a comparison, detecting if a term is flexible cannot be done reliably in  $\lambda$ Prolog.

An interpreter for the language can take the meta-level metaphor literally: we implemented constraint propagation by starting a new interpreter that takes in input the guard of the propagation rule and a reflection of the syntax of the goals of the interpreter below. In principle, we could even delay goals in the meta-level and propagate them using a meta-meta-level.

The bijection described by the alignment expression is key to manipulate in the same guard (or combine in the new goal) terms living in different contexts.

As an implementation detail, heads are *matched* against the constraints by the interpreter, and only when matching and alignment succeed, freezing takes place, the meta-interpreter is started and the guard is run.

When the meta-interpreter halts successfully on the guard, the new goal is defrost and it is immediately scheduled for execution in the interpreter. Defrosting restores the metavariables originally present in the constraints, and it also creates a new metavariable (in the interpreter) for each new metavariable (in the meta-interpreter) that was generated by executing the guard.

For example, if we take the two typing constraints on  $P$  above and we freeze them, we obtain:

```
of x (sort c0) x ?- of (c1 x) (arr x _\ x) (c1 x)
of z (sort c0) z ?- of (c1 z) (arr (c2 z) y\ c3 z y) (c1 z)
```

Note that equal metavariables are frozen with the same fresh constant. Eigenvariables are left untouched by freezing, it is the work of the alignment phase to make sense of them. In this simple case there is only one possible bijection between the two (singleton) sets of eigenvariables:  $x \mapsto z$ .

The following query is run in the meta-interpreter to validate the guard of the rule.

```
?- equiv [of z (sort c0) x] [of z (sort c0) x]
```

Finally the following goal is generated and scheduled for the next SLD resolution step.

```
of z (sort I) z => conv (arr z _\ z) (arr (T z) y\ S z y)
```

The new goal eventually assigns  $\tau$  and  $s$ , and its success or failure is semantically equivalent to the satisfiability of the second constraint on  $\mathbb{P}$ , that is removed.

#### 4.3. Automatic alignment of goals

Matching of  $\lambda$ Prolog goals is a delicate matter: eigenvariables are fresh names, not fixed constants, and their semantics is quite different. In particular, matching goal  $H$  against goal  $G$  amounts to solving this equation written using the nabla quantifier of (GMN11):  $\nabla x_1 \dots x_n, \mathcal{H} = \nabla y_1 \dots y_m, \mathcal{G}$ .

In the general case it is closely related to equivariant unification that is known to be a NP hard problem.(Che04) While being an elegant high-level language, CHR is not efficient. It is estimated be on average 100 times slower than conventional programming languages. If, by handling higher order constraints, we make its core operation, matching, even more expensive, we are unlikely to obtain a working system.

Provisionally the use case that drove our design seems to be on a lucky spot: all relevant names are visible by the key of the delayed goal. Indeed, in the Curry-Howard homomorphism, a metavariable represents a sequent, and its proof (and type) can only use variables in scope of the metavariable. The design choice we make is to forbid the  $\nabla$  quantifier in the patterns of CHR rules, and implicitly consider as many  $\nabla$  quantifications as eigenvariables visible to the key variable of the constraint being matched by the CHR head. This way we can impose the trivial bijection between the names in the pattern and the names in the constraint. Also, all patterns use distinct metavariables, so all injections are trivial. In other words, names are only dealt with in the alignment. We support two form of alignments: one “automatic” based on the  $L_\lambda$  invariant and one “manual” for programs that go outside that fragment.

In  $L_\lambda$  the variables visible by each constraint key are distinct, and equating the (same) keys gives a bijection between the names. If a CHR rule has  $n$  patterns, and each corresponding goal has  $m$  eigenvariables, then the alignment is computed in  $O(m^n)$

The manual alignment injects all eigenvariables to the disjoint union of the eigenvariables visible by each constraint key. In other words the terms bound by the CHR rule’s heads share no names: it is up to the programmer to eventually relate the names. In this case we also let the programmer access the list of terms to which the metavariable is applied with the following syntax:  $(?? \ \kappa \ \mathbb{L} \ \text{as } \mathbb{X})$ . Here  $\kappa$  is the (frozen) metavariable,  $\mathbb{L}$  is the list of terms to which it is applied and  $\mathbb{X}$  is just  $\kappa$  applied to  $\mathbb{L}$ . We put this feature to good use in the next section.

#### 4.4. Meta level: elegant but overkill

The operational semantics of the rule  $S_1 \setminus S_2 \mid G \iff S_3$  replaces  $S_2$  with  $S_3$ . Therefore its soundness only requires  $\bigwedge S_1 \wedge G \Rightarrow (\bigwedge S_2 \Leftarrow \bigwedge S_3)$ . When it is not the case that also  $\bigwedge S_1 \wedge G \Rightarrow (\bigwedge S_2 \Rightarrow \bigwedge S_3)$ , the rewriting rule is not complete or, equivalently, the user is only specifying and heuristic that potentially throws away solutions.

Heuristics are something the user surely wants to employ when extending the elaboration

tor. Our running example is the following emblematic conversion (or, better, narrowing) problem

```
conv (app carrier_of_group R) integers
```

In CIC a term can pack together types, terms and properties. In the example above `R` of type `group` is to be instantiated with a record that packs together the group carrier, the group operations and their properties. In this setting one can clearly see that the generative mode of  $\lambda$ Prolog corresponds to proof search, i.e. the blind enumeration of  $\lambda$ -terms to build (uninteresting) groups over the integers. Of course the user is likely to have already built, in her library, the standard ring of integers, and he would like to instruct the system with the heuristics that picks for `R` that group. The code to do so is the first propagation rule below:

```
delay (conv (app carrier_of_group X) Y) on X.
constraint conv {
  rule \ (conv (carrier_of_group X) integers) <=> (X = integer_group).
  rule \ (conv (carrier_of_group X) (intersection A B)) <=>
    (conv (carrier_of_group GA) A, conv (carrier_of_group GB) B,
     X = intersection_group GA GB).
}
```

The second rule is recursive and re-phrases the theorem saying that the intersection of two groups is a group: in order to find a group `x` whose carrier is the intersection of the two sets `A` and `B`, the heuristic suggests to recursively discover two groups `GA` and `GB` of carrier `A` and `B`, and to instantiate `x` with the group intersection of `GA` and `GB`.

Remark how little we use of CHR in this rule: only one head, no alignment, no guard (i.e. no need to freeze). Still, these simple rules faithfully model the extension to the elaborator of Coq that proved to be key to outstanding formalizations like the Odd Order Theorem (GAA<sup>+</sup>13). Such formalization declares a 26 hundreds of rules like these ones. It goes without saying that the efficiency of such rules is critical.

To achieve efficiency, in this simple scenario we would like to avoid delaying the goal, matching it with a propagation rule, execute the guard at the meta-level, etc. To do so, we will expose in the language the low level primitives that are used in the interpreter to implement constraint generation and propagation. The idea is that, in the restricted scenario above, the user will be able to directly write the propagation rule in the interpreter.

The first primitive is called `mode` and lets one tag arguments of a predicate as input ones.

```
mode (of i o o). % first argument is input, the other two output
```

The result is that arguments marked as input are *matched* against the corresponding arguments in the goal, exactly as the CHR engine matches the heads of a rule against constraints (without instantiating any metavariable occurring in the constraints). This let us write programs like the following one without risking that the first two rules “generate” the input. Of course the right hand side of the clauses can instantiate metavariables.

```
of (lam T F) T (lam RT RF) :- ...
of (app M N) T (app RM RN) :- ...
of X T RX :- var X, delay (of X T RX) [X].
```

The last rule is the actual implementation of the global directive `delay .. on ..` we described before. Note that this way the condition to identify goals to delay can be made arbitrarily sophisticated. Also remark that the mode declaration for `of` is not equivalent to place a clause like `(of X T RX :- var A, !, ...)` on top, since hypothetical clauses may be placed by the runtime before this one, and they can be generative as well.

The syntax `(?? K L as X)` seen in 4.3 is also available, and lets one access the arguments of a flexible term. This lets one extend the conversion predicate (unification in the elaborator) to cover terms outside  $L_\lambda$ . An emblematic example is the following one:

```
rule (G ?- conv (?? F [nat_of_ord X]) T)
| ((mem (val X _ _ _) G ; mem (of X _ _ _) G),           % X is a bound variable,
   (pi y\ copy (nat_of_ord X) y => copy T (T1 y)),      % T1 (nat_of_ord X) = T,
   (pi w\ copy (T1 X) (T1 w)))                          % X does not occur in T1.
<=> (F = T1)
```

When `x` never occurs in  $T$  naked, but only inside the `nat_of_ord` context,  $T_1$  is  $T$  where all occurrences of `nat_of_ord x` are bound. Such example occurs very frequently in the library of big operators of Coq (BGOBP08), where theorems about iterated operations over finite domains are provided. For example the  $\sum_{i < 7} (F i)$  expression hides the `nat_of_ord` injection around  $i$  (a term of the type of  $I_7$ , a finite subset of the integers of cardinality 7). The impossibility to extend the elaborator of Coq with any heuristic to solve the goal above makes the use of these lemmas painful, since one may have to provide `F` by hand, instead of having it automatically inferred.

In Section 5 we use the low level primitives `mode`, `delay` and `var` to implement our elaborator. In some cases, we combine `mode` and `delay` just to implement the high-level `delay` construct. In some other cases we use `mode` and pattern matching over the syntax of goals to implement heuristics without triggering delay.

In all cases, the low level primitives above can be understood in terms of high-level constraint propagation rule. In particular, semantically `mode` already acts as the request of delaying the goal when certain inputs are flexible. Then, in the goals that match metavariables, one can either confirm the intention of delaying (via `delay`), or he can provide a propagation rule by immediately issuing a new query to be executed by the interpreter.

In the future we plan to study the elaboration of the syntax based on the low level primitives to the high-level syntax and its clean declarative semantics.

#### 4.5. ELPI = $\lambda$ Prolog + CHR

We implemented the language described in the previous sections in an efficient interpreter, written in OCaml, that we called ELPI (Embedded Lambda Prolog Interpreter) and that is open source and downloadable from <https://github.com/LPCIC/elpi>. In addition to the new programming constructs that deal with constraints, ELPI slightly differs from the version of  $\lambda$ Prolog implemented in Teyjus in a few minor aspects.

First, in ELPI all types and type and sort declarations can be omitted, whereas they are mandatory in Teyjus. Originally, and according to (BBR98), types were necessary to implement Huet's algorithm for higher order unification. However, for several years

now the unification algorithm of Teyjus only solves equations in the pattern fragment  $L_\lambda$  discovered by Miller (Mil91), which admits most general unifiers and reduce unification to a decidable problem. All other equations are automatically delayed by Teyjus to be fired only when the equation is instantiated to one in the pattern fragment. At the level of the implementation, the code of ELPI that implements this delay is shared with the one to delay arbitrary predicates.

Second, the module system of Teyjus is not implemented. Only the `accumulate` directive is honored for backward compatibility and with a different semantics. In ELPI we provide instead explicit existentially quantified local constants whose scope spans over a set of program clauses. This mechanism gives in a simple way predicate and constructor hiding that are provided differently by the module system of Teyjus.

Last, in a few corner cases, the parsing of an expression by Teyjus is influenced by the types. In particular, types are used to disambiguate between lists of elements and a singleton list of conjunctions. The syntax is disambiguated in a different way in ELPI.

Despite the differences above, we tried very hard to maintain backward compatibility with Teyjus and its standard library. Indeed, we are able to execute in ELPI all the code from Teyjus that we collected from the Web, up to a very few minor changes to the source code. ELPI has also been validated on the large code base of the Foundational Proof Certificate (RB17) framework.

Last, ELPI is a pure interpreter written in OCaml. Embedding it into larger applications like Coq or Matita is easy (no external program to run, no compilation chain).

## 5. Towards an elaborator for CIC

We are developing an elaborator for CIC as a modular extension of the kernel described in Sect. 2. The elaborator mimics as close as possible the behaviour of the one of Matita 0.99.1 (ARCT12), with the exception of the handling of universe constraints that follows Coq (Her05) and the handling of flexible-flexible narrowing cases that are delayed in our new implementation and that are instead resolved using heuristics in Matita.

In place of following the algorithm of Matita and Coq, an alternative very promising choice would have been to mimic the elaborator described in (MA16) and already presented via typing rules that yield a set of higher order unification constraints to be solved later. The choice to be closer to Matita allow us to easily compare the performances of the elaborator written in ELPI with the one of Matita written in OCaml, in order to further optimism the ELPI interpreter.

To implement the elaborator, we need to consider all the predicates defined by induction over the shape of terms, and either turn them into constraints to be propagated, or immediately suggest solutions.

### 5.1. Type-checking: the `of` predicate

Using a mode declaration plus `delay`, we delay type-checking a metavariable, turning it into a proof obligation, i.e. a sequent of the form  $G \text{ ?- } \text{of } x \text{ t } R_x$  where  $G$  holds type declarations for variables (`of`  $x \text{ t } x$ ) or value assignments (`val`  $x \text{ t } v \text{ nf}$ ).

```

mode (of i o o).
of (?? as K) T RK :- !, delay (of K T RK) K.

```

We then declare one constraint propagation rule that corresponds to a special case of uniqueness of typing in the case of dependently typed languages: one of the two obligations is *canonical*, i.e. it is of the form  $G \text{ ?- of } (X \ x_1 \ \dots \ x_n) \ t \ \_$  where all the  $x_i$  are distinct variables or, equivalently, when  $x \ x_1 \ \dots \ x_n$  is in the pattern fragment  $L_\lambda$  discovered by Miller.

The restriction may seem severe, but, once a canonical typing constraint enters the set of delayed goals, it is simple to reduce the whole set to just the canonical one plus additional conversion constraints. Moreover, the first time a metavariable is generated in an interactive prover like Matita its typing constraint is canonical by construction. Therefore, it is customary both in theory (GJ02) and implementation (ARCT12) to always keep only canonical constraints, that are collected in a set called *metavariable environment* (metasenv) and that are the only proof obligations presented to the user.

The propagation works as follow: let  $G_1 \text{ ?- of } (X \ x_1 \ \dots \ x_n) \ u_1 \ \_$  be the canonical sequent and  $G_2 \text{ ?- of } (X \ t_1 \ \dots \ t_n) \ u_2 \ \_$  be the one to be simplified. First we compute the type of  $x_1$  in  $G_1$  and of  $t_1$  in  $G_2$ , imposing as a new constraint their convertibility in the union of  $G_1$  and  $G_2$ . Then we proceed recursively over  $x_2 \ \dots \ x_n$  and  $t_2 \ \dots \ t_n$  after assigning  $t_1$  to  $x_1$  via **val** in  $G_1$ . When the two lists become empty, we generate a final constraint to check if  $u_1$  is convertible with  $u_2$ .

To detect canonicity, we use the ad-hoc extension predicate **name** of ELPI that holds iff the argument is a universal variable. The syntax  $G \Rightarrow L$  when  $G$  is a list of propositions is equivalent in ELPI to assuming the conjunction of the predicates in  $G$ .

```

constraint of val {
  rule (G1 ?- of (?? _ L1 as X1) T1 _) \ (G2 ?- of (?? _ L2 as X2) T2 _) > X1 ~ X2
  | (is_canonical L1, compat G1 L1 T1 G2 L2 T2 L3) <=> L3.
}

is_canonical [].
is_canonical [X|XS] :- name X, not (mem X XS), is_canonical XS.

% (C ?- TC) is the canonical one
compat C [] TC L [] TL H :- append C L CTX, H = (CTX => conv TL TC).

compat C [V|VS] TC L [Arg|Args] TL K :-
  H1 = (L => of3 Arg U1 R1),
  H2 = (C => of3 V U2 R2),
  append C L CTX, H3 = (CTX => conv U1 U2),
  compat [val V U2 R1 _NF|C] VS TC L Args TL K2,
  K = (H1, H2, H3, K2).

```

Last, remark that the alignment mode selected only checks that  $x_1$  and  $x_2$  are the same metavariable and makes all eigenvariables distinct.

## 5.2. Universe constraints: the *lt*, *succ*, *max* predicates

All three predicates must be delayed when at least one of the arguments is flexible. However, satisfiability of the constraints is a necessary requirement for logical consistency.

In case of violation of the constraints, it is indeed easy to encode a form of Russel’s paradox.

In order to verify satisfiability, we could devise a set of propagation rules for constraints over integers induced by the last three predicates. However, to preserve the logical consistency of the system, it is not necessary to keep the total, discrete order of integers. On the contrary, it is more flexible for the user to assume a generic partially ordered set  $(\text{univt}, \text{lteq})$ , and to relax the successor relation to being strictly greater and the max to a generic upper bound. Satisfiability of the set is now equivalent to the absence of a strict cycle, i.e. to the possibility to derive  $\text{ltn } \cup \cup$  for some universe  $\cup$ .

Detecting an inconsistency from a set of constraints expressed using strict and lax inequalities is such an easy job for CHR that we basically just had to modify the “hello world” example for CHR given on Wikipedia, that simplifies constraints over lax inequalities only. Each constraint propagation rule corresponds to an instance of a characterizing property of the order, i.e. reflexivity, transitivity and antisymmetry of **leq**, transitivity and irreflexivity of **ltn**, and finally inconsistency of **ltn**  $\times$   $\forall$  with both **ltn**  $\forall$   $\times$  and **leq**  $\forall$   $\times$ . We only show in the following code a few propagation rules.

```

kind univt type.
macro @univ :- univt.

lt I J :- ltn I J.
succ I J :- ltn I J.           % succ relaxed to <
max N M X :- leq N X, leq M X. % max relaxed to any upper bound

... /* boilerplate code to delay leq and ltn over flexible terms */

constraint leq ltn {
  % incompatibility and irreflexivity
  rule (leq X Y) (ltn Y X) <=> fail. rule (ltn X X) <=> fail.
  % reflexivity, antisymmetry, ...
  rule \ (leq X X). rule (leq X Y) \ (leq Y X) <=> (Y = X). ...
}

```

The code just shown is sufficient to infer universe levels for the predicative fragment of ECC. However, the ECC implemented by Matita has also a sort `prop` for impredicative propositions and a mirror copy of the predicative hierarchy to differentiate between universes of data types **type** `lv1` and universes of predicative propositions **cprop** `lv1`. Therefore the actual code implemented instantiates `@univ` with all three kind of universes and implements the **lt**, **succ**, **max** predicates according to the PTS of Matita. In particular, to compare predicative universes the code boils down to calls to the **ltn** and **leq** predicates implemented above using CHR-style propagation rules.

### 5.3. Reduction: the `whd1`, `whd*` predicates

The predicate **whd\***, defined in terms of **whd1**, computes the normal form of the input. It is used by **match\_sort** and **match\_arr** to verify if the normal form has a given shape. Moreover, it calls itself recursively to compute the normal form of arguments according to the call-by-need strategy.

The typing system we are implementing cannot distinguish between a term and its

normal form. Therefore, when computing the normal form of a flexible input  $x$ , it is meaningless to delay a goal stating that  $y$  is the normal form of  $x$ , because typing does not distinguish them. Therefore, we just return  $x$  as the normal form of  $x$  by letting `whd1` fail over flexible terms.

```
mode (whd1 i i o o).
whd1 ?? _ _ _ :- fail.
```

The consequence on the strategy is that, under certain alternations of reductions and instantiations of metavariables, the implemented strategy will be intermediate between call-by-name and call-by-need, recomputing the normal form of arguments that were metavariables at the time of their first use, with no consequences on typability.

On `match_sort` and `match_arr` there is no consequence: the metavariable is immediately assigned the wanted shape, meaning that we have decided to instantiate the metavariable immediately with its normal form.

#### 5.4. Term comparison: the *sub*, *conv* predicates

Conversion when at least one of the arguments is a flexible term amounts to higher order narrowing. Instead of delaying the goal, an heuristic already used in Matita immediately rewrites the constraint by solving the unification problem. The heuristic always favors projections to mimic (in Huet's terminology). Further heuristics are used in Matita to avoid projecting over flexible arguments, to make unification more predictable to the user. These heuristics are clearly incomplete, discarding all solutions but one and sometimes failing to find a solution when it exists. However, they are more general than the ones implemented in Coq and, from a practical perspective, they guess most of the time the unifier that is expected by the user when interacting with the ITP.

The following code shows a simplified version of the clauses that deal with a flexible term (on the left-hand side) to be unified with a rigid one on the right-hand side. A symmetric set of rules is required for the dual case, and the code can obviously be unified with minor effort. Flexible-flexible cases are instead detected and delayed, whereas Matita and Coq have complex heuristics to solve them as well.

```
mode (comp i i i i i).

% Delay flex-flex cases
comp (?? as V1) S1 M (?? as V2) S2 :- !, delay ( comp V1 S1 M V2 S2) [V1,V2].

% T1 :- lam TYA F + beta step
comp (?? as T1) [A|AS] M T2 L2 :-
  of A TYA RA, T1 = lam TYA F, pi x \ val x TYA RA _NF => comp (F x) AS M T2 L2.

% Auxiliary predicates for projection heuristics
on_rigid_term B S CONT :- whd* B S (_ \ B2 \ S2 \ not (var B2), CONT B2 S2).
is_rigid_term B S :- on_rigid_term F S (_ \ _ \ true).
eta (lam _ F) [] :- pi x \ is_rigid_term (F x) [].

% PROJECTION
comp (?? as V1) [] M T2 S2 :-
  on_rigid_term T2 S2 (T2b \ S2b \ not (eta T2b S2b), val X XT B BN, V1 = X,
  on_rigid_term B [] (Bb \ Sb \ comp Bb Sb M T2b S2b)), !.
```

```

% MIMIC
% non empty stack = application
comp (?? as V1) [] _ T2 S2 :-
  append S21 [S21] S2, occur_check S21 V1,
  V1 = app X Y, comp X [] eq T2 S21, comp Y [] eq S21 [].
% regular mimic rules
comp (?? as V1) [] M T2 [] :- occur_check T2 V1, mcomp V1 M T2. % mcomp used generatively
% variables and constants
comp (?? as V1) [] M T2 [] :- is_atom T2, V1 = T2, !.

% REDUCTION (same rule as the kernel)
comp (?? as V1) [] D T2 S2 :- whd1 T2 S2 (_ \ comp V1 [] D).

% FAIL
comp ?? [] _ _ _ :- !, fail. % to avoid the fast path of the kernel

% MIMIC RULES (same rules as the kernel)
mcomp (sort I) eq (sort I) :- !.
mcomp (sort J) leq (sort I) :- !, leq J I.
mcomp (app A1 B1) _ (app A2 B2) :- !, comp A1 [] eq A2 [], comp B1 [] eq B2 [].
mcomp (lam TY1 F1) _ (lam TY2 F2) :-
  !, comp TY1 [] eq TY2 [], (pi x \ comp (F1 x) [] eq (F2 x) []).
mcomp (arr TY1 F1) M (arr TY2 F2) :-
  !, comp TY1 [] eq TY2 [], (pi x \ comp (F1 x) [] M (F2 x) []).

```

Note that the mimic steps are attempted only after a call to an occur check predicate (not shown in the paper, but simply implemented recursively).

To perform projection steps, the code looks for the first visible constant that is bound to a term that is the one the metavariable needs to be unified with. For example, the solution to the unification problem `comp (X x) [] M t []` where `val x _ t2 _` is an assumption and `t` and `t2` are unifiable is solved by instantiating `x` with the identity (projection step). In order to avoid wild guesses and potential divergence leading to infinite eta-expansions, a projection step is only attempted if both `t` and `t2` are rigid. These heuristic guards are already found in the code of Matita.

### 5.5. Automatic saturation

Matita was the first prover to allow in the syntax both placeholders for an omitted term and placeholders for a possibly empty sequence of omitted terms. The concrete syntax for the latter placeholders is “...”. For example, Matita accepts `eq_f2 ... plus` and elaborates it to `eq_f2 nat nat nat plus` where `eq_f2` is a constant that expects in input three types  $A, B, C$  and a function of type  $A \Rightarrow B \Rightarrow C$ .

The operation that fills in the dots is called *saturation* and the semantics of the dots is greedy: the dots can only be used in argument position in an application and they are turned in the minimal sequence of inferred terms that make that application type-check. The code of many tactics is highly simplified by automatic saturation: it is sufficient for the tactic writer to insert enough dots here and there and let the elaborator figure out the number of omitted arguments and their instantiation. Dots also reduce the need for implicit arguments in the prover syntax.

In our elaborator we use the syntax `hole` for a placeholder for a single term and `vect` for the dots. The code that handles them only requires a few lines:

```

type hole,vect term.
% turn an hold into a metavariable and repeat
of hole U RT :- !, of T U RT.
% try first to turn dots into zero holes
of (app M vect) TM RM :- of M TM RM.
% otherwise add one more hole and repeat
of (app M vect) U RT :- !,
  of M TM RM, not (var TM), match_arr TM _ _, of3 (app (app RM hole) vect) U RT.

```

### 5.6. Coverage of the full type theory of Matita

We have modularly extended the kernel presented in Section 2.5.1 to cover all the rules of the elaborator of Matita, except for the flexible-flexible case of unification that we currently delay. We then branched the code to Matita in order to compare the behaviour and performance of the two implementations.

The code is still being debugged and can be found here: <https://github.com/LPCIC/matita>. Currently the majority of the elaboration problems invoked by Matita compiling its arithmetic library are solved by the  $\lambda$ Prolog code as well. Minor bugs could be responsible for a few of the ones that do not pass. The majority of the problems that do not pass are due to the different heuristics implemented, in particular for the flex-flex case. For example, by currently delaying those unification problems we also end-up delaying occur checks that force Matita to look for different solutions. Indeed, at least several failing problems in  $\lambda$ Prolog time-out after a while when diverging due to undetected occur-checks.

In the future we plan to bring the heuristics of the two implementations full in sync. Because they are currently not in sync, it is hard to compare the performances of the two systems. Currently on the majority of examples that pass the elaborator is as slow as the kernel, i.e. by one order of magnitudes. However, we also have fewer examples that are two order of magnitudes slower. We are investigating these cases one by one, and it usually boils down to a difference in the heuristics (the ones of Matita has been carefully improved for years).

From the point of view of code size, instead, the situation is very good: only 323 lines of ELPI are required to turn the kernel in an elaborator, whereas the implementation of the elaborator in Matita requires about 2.500 lines.

We notice however that, due to automatic backtracking, debugging the ELPI version is harder when the code follows an unexpected path. In the future we will need to understand how to improve trace-ability of the code.

## 6. Conclusions and related works

In this paper we validate  $\lambda$ Prolog as a good language to implement the type checker (kernel) of a fully fledged type theory like the Calculus of Inductive Constructions. In order to implement an elaborator, i.e. a type checker for partial terms, we extend the

$\lambda$ Prolog with constructs to turn goals into constraints and we add a CHR like language to manipulate the set of constraints. We implement the proposed extensions in the ELPI system and use them to extend, in a fully modular way, the kernel into an elaborator.

To our knowledge ELPI is the first  $\lambda$ Prolog implementation extended with first class constraints. The situation is very different for Prolog, where all mainstream implementations integrate some constraint solvers. Moreover, the CHR language is typically compiled to Prolog, hence Prolog systems can quite easily provide a CHR package. It is for example the case for SWI Prolog (WSTL12) and SICStus Prolog (AAB+93). Providing to the Prolog language primitives of the meta-level, like matching, has also been tried before. For example the Picat language, based on the B-Prolog (Zho12) engine, lets the user chose, for each clause, if the head has to be matched or unified with the goal.

While basing the implementation of the kernel of an interactive prover on a logical framework (LF) eventually animated by SLD resolution is not new, little has been done to reuse the same technology for the elaborator component. For example MMT (Rab13) “meta” system lets one define kernel rules in LF, but resorts to the Scala language for the elaborator. Isabelle lets one describe the axioms of a logic in an LF framework and exposes the higher order unification algorithm to the higher layers of the system, like the proof language one. Incidentally the dominant logic implemented in Isabelle is HOL, that lacks dependent types, hence needs no term comparison algorithm based on narrowing. As a consequence, no need to extend such algorithm to encompass more rewriting is perceived, and no way to extend such algorithm is given to the user. The Dudukti (BCH12) language lets one describe the axioms of a logic in LF modulo equational theories. Unfortunately, at the time of writing, the system lacks an elaborator.

Describing the elaborator in terms of constraints was a choice first made in Agda version 2 (Nor09) and more recently in Lean (dMAKR15). Both system implement their own, ad-hoc, constraint solver, the former system in Haskell, while the latter in C++. The approach we follow in the paper is to provide a programming platform where constraints and their propagation rules are first class, to ease documentation, experimentation and extension of the implemented system.

A big advantage of writing the code in  $\lambda$ Prolog is the possibility of using the Abella (Gac08) theorem prover to mechanically check its correctness. In particular, Abella can be used to prove the soundness of all constraint propagation rules. For example, for an early prototype of the elaborator presented in this paper, the typing constraint propagation rule implementing uniqueness of typing was proved correct in Abella. However, the current version of Abella reasons on the big step operational semantics of  $\lambda$ Prolog programs without cuts. Because delaying a goal means interrupting execution in the middle, the small step version of the operational semantics would be the appropriate tool to use instead. For example, in the aforementioned proof of uniqueness of typing, we had to assume a certain invariant on the  $\lambda$ Prolog context that is the context of the delayed typing sequent. In order to know that the invariant really holds during computation, we need to prove that, if it holds initially, it holds after any number of small execution steps. Currently, this statement cannot even be stated in Abella. In the future, we would like to re-design Abella around a small step operational semantics to overcome this limitation and also to be able to reason over cuts.

We plan to compare in a disciplined way the performance of the elaborator of Matita with the one presented in this paper, to precisely size the trade-off between the programming abstractions offered by the HOCLP language we have implemented and their runtime cost.

As another future work, we hope in Abella to fully prove correct, although not complete, the elaborator component, so to call into question the standard design of interactive provers: what is the point of having a kernel component if the elaborator is proved to only generate terms accepted by it?

## References

- Johan Andersson, Stefan Andersson, Kent Boortz, Mats Carlsson, Hans Nilsson, Thomas Sjöland, and Johan Widen. SICStus Prolog User’s Manual. Technical report, 1993.
- Beniamino Accattoli, Pablo Barenbaum, and Damiano Mazza. Distilling abstract machines. In *Proceedings of the 19th International conference on Functional programming*, pages 363–376, 2014.
- Andrea Asperti, Wilmer Ricciotti, Claudio Sacerdoti Coen, and Enrico Tassi. Hints in Unification. In *Theorem Proving in Higher Order Logics, 22nd International Conference, TPHOLs 2009, Munich, Germany, August 17-20, 2009. Proceedings*, pages 84–98, 2009.
- Andrea Asperti, Wilmer Ricciotti, Claudio Sacerdoti Coen, and Enrico Tassi. A Bi-Directional Refinement Algorithm for the Calculus of (Co)Inductive Constructions. *Logical Methods in Computer Science*, 8(1), 2012.
- A. Asperti, W. Ricciotti, C. Sacerdoti Coen, and E. Tassi. A compact kernel for the Calculus of Inductive Constructions. *Sadhana*, 34(1):71–144, 2009.
- Andrea Asperti, Wilmer Ricciotti, Claudio Sacerdoti Coen, and Enrico Tassi. *The Matita Interactive Theorem Prover*, pages 64–69. Springer Berlin Heidelberg, Berlin, Heidelberg, 2011.
- Bruno Barras. *Auto-validation d’un système de preuves avec familles inductives*. Thèse de doctorat, Université Paris 7, November 1999.
- Catherine Belleanne, Pascal Brisset, and Olivier Ridoux. A Pragmatic Reconstruction of Lambda-Prolog. In *Journal of Logic Programming*, 1998.
- Mathieu Boespflug, Quentin Carbonneaux, and Olivier Hermant. The lambda-Pi-calculus Modulo as a Universal Proof Language. In *Proceedings of the Second International Workshop on Proof Exchange for Theorem Proving*, volume 878 of *CEUR Workshop Proceedings*, pages 28–43, 2012.
- Yves Bertot, Georges Gonthier, Sidi Ould Biha, and Ioana Pasca. Canonical Big Operators. In *Theorem Proving in Higher Order Logics*, volume 5170/2008 of *LNCS*, 2008.
- James Cheney. *The Complexity of Equivariant Unification*, pages 332–344. Springer Berlin Heidelberg, 2004.
- Claudio Sacerdoti Coen and Enrico Tassi. Nonuniform Coercions via Unification Hints. In *TYPES*, pages 16–29, 2009.
- Cvetan Dunchev, Claudio Sacerdoti Coen, and Enrico Tassi. Implementing HOL in an Higher Order Logic Programming Language. In *Proceedings of the Eleventh Workshop on Logical Frameworks and Meta-Languages: Theory and Practice, LFMTP ’16*, pages 4:1–4:10. ACM, 2016.
- Cvetan Dunchev, Ferruccio Guidi, Claudio Sacerdoti Coen, and Enrico Tassi. ELPI: fast, Embeddable,  $\lambda$ Prolog Interpreter. In *Proceedings of LPAR*, 2015.

- Leonardo Mendonça de Moura, Jeremy Avigad, Soonho Kong, and Cody Roux. Elaboration in Dependent Type Theory. *CoRR*, abs/1505.04324, 2015.
- Leonardo de Moura, Soonho Kong, Jeremy Avigad, Floris van Doorn, and Jakob von Raumer. *The Lean Theorem Prover (System Description)*, pages 378–388. Springer International Publishing, Cham, 2015.
- The Coq development team. *The Coq proof assistant reference manual*. Inria, 2017. Version 8.6.
- Thom Frhworth. *Constraint Handling Rules*. Cambridge University Press, 1st edition, 2009.
- Georges Gonthier, Andrea Asperti, Jeremy Avigad, Yves Bertot, Cyril Cohen, François Garillot, Stéphane Le Roux, Assia Mahboubi, Russell O’Connor, Sidi Ould Biha, Ioana Pasca, Laurence Rideau, Alexey Solovyev, Enrico Tassi, and Laurent Théry. A Machine-Checked Proof of the Odd Order Theorem. In *ITP*, volume 7998 of *LNCS*, pages 163–179. Springer, 2013.
- Andrew Gacek. The Abella Interactive Theorem Prover (System Description). In *Proceedings of IJCAR 2008*, volume 5195 of *LNAI*, pages 154–161. Springer, 2008.
- Herman Geuvers and Gueorgui I. Jojgov. Open Proofs and Open Terms: A Basis for Interactive Logic. In *Computer Science Logic, 16th International Workshop, CSL 2002, 11th Annual Conference of the EACSL, Edinburgh, Scotland, UK, September 22-25, 2002, Proceedings*, pages 537–552, 2002.
- Andrew Gacek, Dale Miller, and Gopalan Nadathur. Nominal abstraction. *Information and Computation*, 209(1):48 – 73, 2011.
- Hugo Herbelin. Type inference with algebraic universes in the Calculus of Inductive Constructions. Available at <http://pauillac.inria.fr/~herbelin/articles/univalgccii.pdf>, 2005.
- Jean-Louis Krivine. A Call-by-name Lambda-calculus Machine. *Higher Order Symbol. Comput.*, 20(3):199–207, 2007.
- Z. Luo, S. Soloviev, and T. Xue. Coercive subtyping: Theory and implementation. *Inf. Comput.*, 223:18–42, February 2013.
- Zhaohui Luo. ECC, an Extended Calculus of Constructions. In *Proceedings of the Fourth Annual Symposium on Logic in Computer Science (LICS ’89), Pacific Grove, California, USA, June 5-8, 1989*, pages 386–395, 1989.
- Zhaohui Luo. Coercive Subtyping in Type Theory. In *Computer Science Logic, 10th International Workshop, CSL ’96, Annual Conference of the EACSL, Utrecht, The Netherlands, September 21-27, 1996, Selected Papers*, pages 276–296, 1996.
- Francesco Mazzoli and Andreas Abel. Type checking throug unification. *CoRR*, abs/1609.09709, 2016.
- Dale Miller. *A logic programming language with lambda-abstraction, function variables, and simple unification*, pages 253–281. Springer Berlin Heidelberg, 1991.
- Dale Miller. Unification Under a Mixed Prefix. *J. Symb. Comput.*, 14(4):321–358, 1992.
- Dale Miller and Gopalan Nadathur. Higher-order logic programming. In *3rd Int. Conf. Logic Programming*, volume 225 of *LNCS*, pages 448 – 462. Springer-Verlan, 1986.
- Dale Miller and Gopalan Nadathur. *Programming with Higher-Order Logic*. Cambridge University Press, 1st edition, 2012.
- Dale Miller, Gopalan Nadathur, Frank Pfenning, and Andre Scedrov. Uniform proofs as a foundation for logic programming. *Annals of Pure and Applied Logic*, 51, 1991.
- Ulf Norell. Dependently Typed Programming in Agda. In *Proceedings of the 6th International Conference on Advanced Functional Programming, AFP’08*, pages 230–266. Springer-Verlag, 2009.
- Christine Paulin-Mohring. *Définitions Inductives en Théorie des Types d’Ordre Supérieur*. Habilitation à diriger les recherches, Université Claude Bernard Lyon I, December 1996.

- F. Rabe. The MMT API: A Generic MKM System. In J. Carette, D. Aspinall, C. Lange, P. Sojka, and W. Windsteiger, editors, *Intelligent Computer Mathematics*, pages 339–343. Springer, 2013.
- Dale Miller Roberto Blanco, Zakaria Chihani. Translating between implicit and explicit versions of proof. In *CADE-26*, 2017.
- Benjamin Werner. *Une Théorie des Constructions Inductives*. PhD thesis, Université Paris VII, May 1994.
- Jan Wielemaker, Tom Schrijvers, Markus Triska, and Torbjörn Lager. SWI-Prolog. *Theory and Practice of Logic Programming*, 12(1-2):67–96, 2012.
- Neng-fa Zhou. The Language Features and Architecture of B-prolog. *Theory Pract. Log. Program.*, 12(1-2):189–218, 2012.